
What is Linearizing?

« o V4
Linear” solvers
+ Linear and convex quadratic objectives & constraints

< Continuous or integer variables (or both)
<+ CPLEX, Gurobi, Xpress, MOSEK, SCIP, CBC, . ..

“Not Linear” problems
< Objectives & constraints in any other form
< Same continuous or integer variables

Goals

< Apply linear solvers to not linear problems
<+ Make this as easy as possible

Example 1: Product of Variables

Can I linearize it?
% ...+cxy+... in my objective

* where x, y are variables; ¢ is a positive constant

It depends . . .

< What kinds of variables are x and y?
< Are you minimizing or maximizing?

Example 1
Case 1: Binary, Minimize

Original formulation

param c > O;
var x binary;
var y binary;

minimize Obj: ... + ¢ * x * y + ...
Linearization

var z;

minimize Obj: ... + ¢c *x z + ...

subject to zODefn: =z >= 0;

subject to zxyDefn: z >=x + y - 1;

...z can be continuous

(minimization forces it to 0 or 1)

Example 1

Case 1 (cont'd)

Many other reformulations possible
+ Best choice depends on problem and solver

< This one seems the best overall choice

* see tests in Jared Erickson’s dissertation:
JaredErickson20120u.northwestern.edu

Extends to product of two linear terms
< Multiply them out

Example 1

Case 1 (cont'd)

General model . . .

param n > O;
param c {1..n} > 0;
param d {1..n} > 0;

var X {1..n} binary;
var Y {1..n} binary;

minimize 0Obj:
(sum {j in 1..n} c[jI*X[j]) * (sum {j in 1..n} d[jI*Y[jl);

subject to SumX: sum {j in 1..n} j * X[j] >= 2*n+3;
subject to SumY: sum {j in 1..n} j * Y[j] >= 2*n+3;

subject to SumXY: sum {j in 1..n} (X[j] + Y[jl) = n;

Example 1

Case 1 (cont'd)

Transformed automatically
<« AMPL interface . . .

% multiplies out the linear objective terms
% sends quadratic coefficient list to solver

< Solver. ..
% transforms products of binaries to linear formulations

Solved . . .

ampl: model xyl.mod;
ampl: option solver cplex;

ampl: solve;

CPLEX 12.6.0.0: optimal integer solution; objective 232.6083992
395 MIP simplex iterations
0 branch-and-bound nodes

Example 1
Case 2: Binary, Maximize

Original formulation

param c > O;
var x binary;
var y binary;

maximize Obj: ... + ¢ * x * y + ...
Linearization

var z;

maximize Obj: ... + ¢c *x z + ...

subject to zxDefn: z <= x;

subject to zyDefn: z <= y;

. .. Z can be continuous
(maximization forces it to 0 or 1)

Example 1

Case 2 (cont'd)

Constraints depend on objective sense
< Minimize: z >= 0, z > x +y - 1

<+ Maximize: z <= x, z <=y

Would it help to include all?

< No, the continuous relaxation is not tightened

% But may need all when
extending this idea to xy in constraints

Example 1
Case 3: Binary & Continuous, Minimize

Original formulation

param c > O;
var x binary;
var y >= L, <= U;

minimize Obj: ... + ¢c * x *x y + ...
Linearization

var z;

minimize Obj: ... + ¢c *x z + ...

subject to zLDefn: z >= L * x;

subject to zUDefn: z >=y - U *x (1 -x);

Example 1

Case 3 (cont'd)

Extends in obvious ways
< Maximization form is symmetric

< y may be integer rather than continuous
* reduces to binary case with [L, U] = [0,1]
Extends to product of two linear terms

+ Equate the y term to a new variable (optional)
< Multiply terms out

Example 1

Case 3 (cont'd)

General model . . .

param n > O;
param c {1..n} > 0;
param d {1..n} > 0;

var X {1..n} binary;
var Y {1..n} >= 0, <= 2;

minimize 0Obj:
(sum {j in 1..n} c[jI*X[j]) * (sum {j in 1..n} d[jI*Y[jl);

subject to SumX: sum {j in 1..n} j * X[j] >= 2*n+3;
subject to SumY: sum {j in 1..n} j * Y[j] >= 2*n+3;

subject to SumXY: sum {j in 1..n} (X[j] + Y[jl) = n;

Example 1

Case 3 (cont'd)

Transformed automatically
<« AMPL interface . . .

% multiplies out the linear objective terms
% sends quadratic coefficient list to Gurobi

+ Gurobi 5.6 solver . ..
% transforms products of variables to linear formulations

Solved by Gurobi 5.6

ampl: model xy3.mod;

ampl: option solver gurobi;

ampl: solve;

Gurobi 5.6.0: optimal solution; objective 177.090486

216 simplex iterations
9 branch-and-cut nodes

Robert Fourer, Strategies for “Not Linear” Optimization
INFORMS Opt Soc Conf — Houston 6-8 March 2014

21




Example 1

Case 3 (cont'd)

Transformed automatically
<« AMPL interface . . .

% multiplies out the linear objective terms
% sends quadratic coefficient list to CPLEX

o CPLEX 12.5 solver. ..

% checks quadratic function for convexity

Rejected by CPLEX 12.5

ampl: model xy3.mod;
ampl: option solver cplex;

ampl: solve;

CPLEX 12.5.0.1: QP Hessian is not positive semi-definite.

Example 1

Case 3 (cont'd)

Transformed automatically

< AMPL interface. ..

% multiplies out the linear objective terms
% sends quadratic coefficient list to CPLEX

¢ CPLEX 12.5.1 solver. ..

% transforms products of variables to linear formulations

Solved by CPLEX 12.5.1

ampl: model xy3.mod;
ampl: option solver cplex;

ampl: solve;

CPLEX 12.5.1.0: optimal integer solution; objective 177.090486
148 MIP simplex iterations

Example 1
Case 4: Continuous, Maximize

Original formulation

param c > O;
var x >= Lx, <= Ux;
var y >= Ly, <= Uy;

maximize Obj: ¢ * x * y;

Conic reformulation

maximize 0Obj: c * z;

subject to zDefn: z72 <= x * y;

Robert Fourer, Strategies for “Not Linear” Optimization
INFORMS Opt Soc Conf — Houston 6-8 March 2014

24




Example 1

Case 4 (cont'd)

Handled by “linear” solvers

< Quadratic conic constraint region is convex
% Original objective was quasi-concave

Can’t sum terms in objective
< Optimal solutions are preserved, but
< Objective value changes (to square root of actual)

% Not a problem if maximizing cx!/?y!/?2

Can't do this with minimize!
< But can minimize a convex quadratic (2xy + x? + y?)
< But can minimize product of negative powers

. . . more with conics in example 3

Example 1

Case 4 (cont'd)

General model . . .

param n > O;
param c {1..n} > 0;
param d {1..n} > 0;

var X {1..n} >= 0, <= 2;
var Y {1..n} >= 0, <= 2;

maximize 0Obj:
(sum {j in 1..n} c[jI*X[j]) * (sum {j in 1..n} d[jI*Y[jl);

subject to SumX: sum {j in 1..n} j * X[j] >= 2*n+3;
subject to SumY: sum {j in 1..n} j * Y[j] >= 2*n+3;

subject to SumXY: sum {j in 1..n} (X[j] + Y[jl) = n;

Example 1

Case 4 (cont'd)

Transformed automatically
<« AMPL interface . . .

% multiplies out the linear objective terms
% sends quadratic coefficient list to Gurobi

% Gurobi solver. ..

% checks quadratic function for convexity

Rejected by Gurobi

ampl: model xy4a.mod;
ampl: option solver gurobi;

ampl: solve;

Gurobi 5.6.0: quadratic objective is not positive definite

Example 1

Case 4 (cont'd)
Model transformed “by hand” . . .

param n > O;

param c {1..n} > 0;
param d {1..n} > 0;

var
var

var
var
var

maximize 0Obj:

subject to ZXdef:
subject to ZYdef:

subject to Zdef: Z72 <= ZX * ZY; # still not positive semidefinite

X {1..n} >=
Y {1..n} >=
ZX >= 0;

ZY >= 0;

Z;

Z;

subject to SumX:

<
<

I
N

0,
0,

I
N

ZX = sum {j in 1..n} c[j1*X[j];
ZY = sum {j in 1..n} d[j1*Y[j];

Example 1

Case 4 (cont'd)

Transformed automatically
<« AMPL interface . . .

% detects quadratic constraint terms
% sends quadratic coefficient list to Gurobi

% Gurobi solver. ..
% detects conic constraint structure

Solved by Gurobi barrier

ampl: model xy4b.mod;
ampl: option solver gurobi;

ampl: solve;

Gurobi 5.6.0: optimal solution; objective 29.78950231
11 barrier iterations

ampl: display ZX*ZY;

ZX*xZY = 887.414 # equals 0bj~2

Example 1

Case 4 (cont'd)

Transformed automatically
<« AMPL interface . . .

% multiplies out the linear objective terms
% sends quadratic coefficient list to CPLEX

Solved by CPLEX branch-and-bound

ampl: model xy4a.mod;

ampl: option solver cplex, cplex_options 'reqconvex 3';

ampl: solve;

CPLEX 12.6.0.0: optimal integer solution; objective 887.4144789
24 MIP simplex iterations

5 branch-and-bound nodes

ampl: option cplex_options 'reqconvex 3 minimize';

ampl: solve;

CPLEX 12.6.0.0: optimal integer solution; objective 88.62097665

67 MIP simplex iterations
15 branch-and-bound nodes

Robert Fourer, Strategies for “Not Linear” Optimization 30
INFORMS Opt Soc Conf — Houston 6-8 March 2014




Example 1
Key Questions

Can it be transformed?
% Yes or no?
¢ Transformed to what?

. . . very sensitive to mathematical form

Who will make the transformation?
< The human modeler?
< The modeling system?
< The solver interface?
< The solver?
. . . Often some combination of these

Example 2: Zero or Range

Can I linearize it?

< If x isn’t zero then I want it to be at least L
* where x > 0 is a variable and L > 0 is a constant

Yes, and . . .

< You can also express it as a discontinuous domain

< You can also express it as a logical condition

Example 2

Scheduling

Minimize number of workers needed
<+ How many workers are assigned to each schedule?

< If a schedule is used at all,
at least L workers must be assigned to it

Data: shifts in each schedule; least assignment L

set SHIFTS;

param Nsched;
set SCHEDS = 1. .Nsched;

set SHIFT_LIST {SCHEDS} within SHIFTS;

param rate {SCHEDS} >= 0;
param required {SHIFTS} >= 0;

param least_assign >= O0;

Example 2

Case 1: Traditional MIP

Zero-one variables and inequalities

var Work {SCHEDS} >= 0 integer;
var Use {SCHEDS} >= 0 binary;

minimize Total_Cost:
sum {j in SCHEDS} ratel[j] * Work[j];

subject to Shift_Needs {i in SHIFTS}:

subject to Least_Usel {j in SCHEDS}:
least_assign * Usel[j] <= Work[j];

subject to Least_Use2 {j in SCHEDS}:

sum {j in SCHEDS: i in SHIFT_LIST[j1} Work[j] >= required[il;

Work[j]l <= (max {i in SHIFT_LIST[jl} required[i]) * Usel[j];

Example 2

Case 1 (cont’d)

Solved by CPLEX

ampl: model schedl.mod;
ampl: data sched.dat;

ampl: option solver cplex;
ampl: let least_assign := 17;

ampl: solve;

Reduced MIP has 269 rows, 252 columns, and 1134 nonzeros.
Reduced MIP has 126 binaries, 126 generals, and O indicators.

Total (root+branch&cut) = 563.38 sec. (138138.56 ticks)

CPLEX 12.6.0.0: optimal integer solution; objective 267
24903192 MIP simplex iterations
3816760 branch-and-bound nodes

Example 2
Case 2: Discontinuous Domain

Union of a point and an interval

var Work {j in SCHEDS} integer in {0} union
interval[least_assign, max {i in SHIFT_LIST[jl} required[il];

minimize Total_Cost:
sum {j in SCHEDS} ratel[j] * Work[j];

subject to Shift_Needs {i in SHIFTS}:
sum {j in SCHEDS: i in SHIFT_LIST[j1} Work[j] >= required[il;

Example 2

Case 2 (cont'd)

Transformed automatically

<+ AMPL processor . . .
% adds auxiliary zero-one variables
% generates appropriate constraints

Solved by CPLEX as usual

Reduced MIP has 269 rows, 252 columns, and 1134 nonzeros.
Reduced MIP has 126 binaries, 126 generals, and O indicators.

Total (root+branch&cut) = 342.49 sec. (85757.81 ticks)

CPLEX 12.6.0.0: optimal integer solution; objective 267
15087185 MIP simplex iterations
2306392 branch-and-bound nodes

Robert Fourer, Strategies for “Not Linear” Optimization
INFORMS Opt Soc Conf — Houston 6-8 March 2014

37




Example 2

Case 2 (cont'd)

Same formulation as case 1

ampl: solexpand;

subject to (Work[1]+IUlb):
Work[1] - 17*(Work[1]+b) >= 0;

subject to (Work[1]+IUub):
-Work[1] + 100*(Work[1]+b) >= 0;

subject to (Work[2]+IUlb):
Work[2] - 17*x(Work[2]+b) >= 0;

subject to (Work[2]+IUub):
-Work[2] + 100*(Work[2]+b) >= 0;

Example 2
Case 3: Implication

CPLEX indicator constraint

var Work {SCHEDS} >= 0 integer;
var Use {SCHEDS} >= 0 binary;

minimize Total_Cost:
sum {j in SCHEDS} ratel[j] * Work[j];

subject to Shift_Needs {i in SHIFTS}:
sum {j in SCHEDS: i in SHIFT_LIST[j1} Work[j] >= required[il;

subject to Least_Use {j in SCHEDS}:
Usel[j]l = 1 ==> Work[j] >= least_assign else Work[j] = 0;

Example 2

Case 3 (cont'd)

Logic passed to solver

+ AMPL writes “logical” constraints as expression trees
<+ AMPL-CPLEX driver “walks” the trees

* detects indicator forms
% converts to CPLEX library calls

Solved by CPLEX with extensions

Reduced MIP has 143 rows, 252 columns, and 882 nonzeros.
Reduced MIP has 126 binaries, 126 generals, and 126 indicators.

Total (root+branch&cut) = 5936.45 sec. (1533625.65 ticks)

CPLEX 12.6.0.0: optimal integer solution; objective 267
250228203 MIP simplex iterations
29437722 branch-and-bound nodes

Example 2
Case 4: Disjunction

Logical constraint using “or” operator

var Work {j in SCHEDS} >= O integer;

minimize Total_Cost:
sum {j in SCHEDS} ratel[j] * Work[j];

subject to Shift_Needs {i in SHIFTS}:
sum {j in SCHEDS: i in SHIFT_LIST[j1} Work[j] >= required[il;

subject to Least_Use {j in SCHEDS}:
Work[j] = 0 or Work[j] >= least_assign;

Example 2

Case 4 (cont'd)

Logic passed to solver

+ AMPL writes “logical” constraints as expression trees
<+ AMPL-CPLEX driver “walks” the trees

% looks for indicator forms

Rejected by CPLEX

ampl: option solver cplex;

ampl: solve;

CPLEX 12.5.0.1: logical constraint not indicator constraint.

Example 2

Case 4 (cont'd)

Logic passed to solver

+ AMPL writes “logical” constraints as expression trees
<+ AMPL-CPLEX driver “walks” the trees

% passes constraints as written to C++ “Concert” interface

Accepted by CPLEX after internal transformation

ampl: option solver ilogcp;

ampl: option ilogcp_options 'optimizer cplex mipdisplay 2';
ampl: solve;

Reduced MIP has 269 rows, 252 columns, and 1134 nonzeros.
Reduced MIP has 126 binaries, 126 generals, and 252 indicators.

<BREAK> (ilogcp)
Total (root+branch&cut) = 95272.30 sec. (23592380.69 ticks)

CPLEX 12.6.0.0: aborted, integer solution exists; objective 267
2.89e+009 MIP simplex iteratioms
351291725 branch-and-bound nodes

Example 3: Ratio

Can I linearize it?
% Minimize Z(l,])EA tl]xl] /T
X Where tl] = bl] + Sijxl'j / (1 — xij/Cij)

* where 0 < x;; < ¢;; is a variable

Yes, but . . .

< The transformation isn’t so obvious
< You will need conic quadratic constraints

Example 3

Traftic Network

Minimize average travel time for given throughput
< How much traffic is routed on each link?
+ Travel time increases nonlinearly with traffic
+ Flow is conserved at nodes

Example 3

Trafftic Network (cont’d)

Model: flows and associated travel times

var Flow {(i,j) in ROADS} >= 0, <= .9999 * capl[i,j];
var Time {ROADS} >= 0;

minimize Avg_Time:
(sum {(i,j) in ROADS} Timel[i,j] * Flowl[i,jl) / through;

subject to Travel_Time {(i,j) in ROADS}:
Time[i,j] = baseli,j] + (sensl[i,jl*Flowli,j]l) / (1-Flowl[i,jl/capli,jl);

subject to Balance_Node {i in INTERS}:
sum{(i,j) in ROADS} Flow[i,j] = sum{(j,i) in ROADS} Flowl[j,il;

subject to Balance_Enter:
sum{ (EN, j) in ROADS} Flow[EN,j] = through;

Example 3

Trafftic Network (cont’d)

Data: Braess network

set INTERS := Db c ;

param EN := a ;

param EX :=d ;

param: ROADS: base cap sens :=
ab 4 10 .1
ac 1 12 .7
cb 2 20 .9
b d 1 15 .5
cd 6 10 i

param through := 20 ;

Example 3

Case 1: General Nonlinear Solver

Solved “locally” by KNITRO

ampl: model traffic.mod;
ampl: data traffic.dat;

ampl: option solver knitro;

ampl: solve;

KNITRO 9.0.0: Locally optimal solution.
objective 61.04695019; feasibility error 1.24e-13
13 iterations; 20 function evaluations

ampl: display Flow, Time;

Flow
ab 9.5b5146
ac 10.4485
b d 11.0044
cb 1.45291
cd 8.99562

Time =
25.2948
57.5709
21.6558

3.41006
14.9564

Example 3

Case 1 (cont'd)

Same with integer-valued variables

var Flow {(i,j) in ROADS} integer >= 0, <= .9999 * capli,jl;

ampl: solve;

KNITRO 9.0.0: Locally optimal solution.
objective 76.26375; integrality gap 1.56e-13
1 nodes; 2 subproblem solves

ampl: display Flow, Time;

Flow Time :=

ab 9 13

ac 11 93.4
bd 11 21.625
cb 2 4

cd 9 15

Example 3
Case 2: “Linear”’ Solver

Rejected by CPLEX

ampl: model traffic.mod;
ampl: data traffic.dat;

ampl: option solver cplex;

ampl: solve;

CPLEX 12.6.0.0:
Constraint _scon[1] is not convex quadratic
since it is an equality constraint.

Example 3

Case 2 (cont'd)

Look at the model again . . .

var Flow {(i,j) in ROADS} >= 0, <= .9999 * capl[i,j];
var Time {ROADS} >= 0;
minimize Avg_Time:
(sum {(i,j) in ROADS} Timel[i,j] * Flowl[i,jl) / through;
subject to Travel_Time {(i,j) in ROADS}:
Time[i,j] = baseli,j] + (sensl[i,jl*Flowl[i,j]l) / (1-Flowl[i,jl/capli,jl);
subject to Balance_Node {i in INTERS}:
sum{(i,j) in ROADS} Flow[i,j] = sum{(j,i) in ROADS} Flowl[j,il;

subject to Balance_Enter:
sum{ (EN, j) in ROADS} Flow[EN,j] = through;

Example 3

Case 2 (cont'd)

Quadpratically constrained reformulation

var Flow {(i,j) in ROADS} >= 0, <= .9999 * capl[i,j];
var Delay {ROADS} >= 0;
minimize Avg_Time:

sum {(i,j) in ROADS} (baseli,jl*Flow[i,j] + Delayli,jl) / through;
subject to Delay_Def {(i,j) in ROADS}:

sens[i,j] * Flow[i,jl"2 <= (1 - Flowli,jl/capli,jl) * Delayli,jl;
subject to Balance_Node {i in INTERS}:

sum{(i,j) in ROADS} Flow[i,j] = sum{(j,i) in ROADS} Flow[j,il;

subject to Balance_Enter:
sum{ (EN, j) in ROADS} Flow[EN,j] = through;

Example 3

Case 2 (cont'd)

Transformed automatically

< AMPL interface. ..

% multiplies out the linear objective terms
% sends quadratic coefficient list to CPLEX

\/
0‘0

Rejected by CPLEX as nonconvex

CPLEX solver. ..

% applies numerical test for elliptic quadratic constraints

ampl:
ampl:

ampl:
ampl:

CPLEX

QP Hessian is not positive

model trafficQUAD.mod;
data traffic.dat;

option solver cplex;
solve;

12.6.0.0:

semi-definite.

Example 3

Case 2 (cont'd)

Quadpratically constrained re-reformulation

var Flow {(i,j) in ROADS} >= 0, <= .9999 * capl[i,j];
var Delay {ROADS} >= 0;
var Slack {ROADS} >= 0;

minimize Avg_Time:
sum {(i,j) in ROADS} (baseli,jl*Flow[i,j] + Delayli,jl) / through;

subject to Delay_Def {(i,j) in ROADS}:
sens[i,j] * Flow[i,j]l~2 <= Slackl[i,j] * Delayli,j];

subject to Slack_Def {(i,j) in ROADS}:
Slack[i,j] = 1 - Flowli,jl/capli,j];

subject to Balance_Node {i in INTERS}:
sum{(i,j) in ROADS} Flow[i,j] = sum{(j,i) in ROADS} Flowl[j,il;

subject to Balance_Enter:
sum{ (EN, j) in ROADS} Flow[EN,j] = through;

Example 3

Case 2 (cont'd)

Transformed automatically
<« AMPL interface . . .

% multiplies out the linear objective terms
% sends quadratic coefficient list to CPLEX

¢ CPLEX solver. ..

* applies symbolic test for conic quadratic constraints

Solved (globally) by CPLEX

ampl: model trafficSO0C.mod;
ampl: data traffic.dat;

ampl: option solver cplex;
ampl: solve;

CPLEX 12.6.0.0: primal optimal; objective 61.04693968

15 barrier iterations

Example 3

Case 2 (cont'd)

Same with integer-valued variables

var Flow {(i,j) in ROADS} integer >= 0, <= .9999 * capli,jl;

ampl: solve;

CPLEX 12.6.0.0: optimal integer solution within mipgap or absmipgap;

objective 76.26375017
19 MIP simplex iterations
0 branch-and-bound nodes

ampl: display Flow;

Flow :=
ab 9
ac 11
b d 11
cb 2
cd 9

Example 3

Case 3

Works just as well with Gurobi

var Flow {(i,j) in ROADS} integer >= 0, <= .9999 * capli,jl;

ampl: solve;

Gurobi 5.6.0: optimal solution; objective 76.26374998
32 simplex iterations

ampl: display Flow;

Flow :=
ab 9
ac 11
b d 11
cb 2
cd 9

Example 3

SOCP-Solvable Forms

Quadpratic
< Constraints (already seen)
< Objectives

SOC-representable
# Quadratic-linear ratios
< Generalized geometric means
< Generalized p-norms

Other objective functions
+ Generalized product-of-powers
+ Logarithmic Chebychev

SOCP-solvable
Quadratic

Standard cone constraints
@ 7i’L=1 a; (fix + gi)z < an+1(fn+1x + gn+1)2r
Ay, 0ny1 =0, £,1X+9p41 =0
Rotated cone constraints

- ?=1 a;(f;x + gi)z < A1 (Fpp1X + gnp1) (Fpg2X + gny2),
ai, - p41 20, £ 91X+ 9n41 20, £0X+ gy 20

Sume-of-squares objectives
< Minimize Z?:l a; (fiX + gi)z

% Minimize v
Subjectto Y a;(fx+g)*<v? v=0

SOCP-solvable

SOC-Representable

Definition
< Function s(x) is SOC-representable iff . . .

* s(x) < a,(f,+1X + gn+1) is equivalent to some
combination of linear and quadratic cone constraints

Minimization property
< Minimize s(x) is SOC-solvable
* Minimize (2
Subject to s(x) < vy 4q

Combination properties
% a-s(x) is SOC-representable for any a = 0
& Yy, 5;(x) is SOC-representable
< max;, s;(x) is SOC-representable
. . . requires a recursive detection algorithm!

SOCP-solvable

SOC-Representable (1)

Vector norm

« ”a ] (FX + g)” — \/Z?:l aiz (fix + gi)z < an+1(fn+1x + gn+1)

% square both sides to get standard SOC
?:1 ai2 (fix + gi)z = a?%+1(fn+1x + gn+1)2

Quadratic-linear ratio

o Z?:l ai(fix + gi)z

< a,.q(f, X+ )
fn+2x+gn+2 n+1\*n+1 In+1

* where f,,,X+ gn42 =0

% multiply by denominator to get rotated SOC
ie1a; (fx+ )% < apyq (B X + Gnyr) (Fi2X + gni)

SOCP-solvable

SOC-Representable (2)

Negative geometric mean
o — H?:l(fix + gi)l/p < fn+1X + 9n+1, P E Z*
% —x;/ 4x;/ 4x31,/ *x;/* < —x5 becomes rotated SOCs:
xg < V1V5, U% < X1X2, Uzz < X3X4

% apply recursively [log, p] times
Generalizations
e —[o (fix + g% < apgp (Fpp X+ gna1): 2ici <1, 4, €QT
e [l (Fix + g) 7% < apyq (Fpp1 X + gna1), @ € QY

% all require f;x + g; to have proper sign

Robert Fourer, Strategies for “Not Linear” Optimization
INFORMS Opt Soc Conf — Houston 6-8 March 2014

68




SOCP-solvable

SOC-Representable (3)

p-norm
(i fix + gilP)? < f01X 4+ gny1, PEQY, p 21

* (Jxq]® + |x,]%)Y5 < x5 can be written

1% /x5 + |x,|° /x5 < x3 which becomes
: 1/5 4 1/5 4
v; + vy, < x5 with —vl/s x3/5 < *xq, —vl/s x3/5 < *x,

* reduces to product of powers

Generalizations
* iz fix + gi|ai)1/a0 <frrX+gne, GEQT, ;= ag =1
2 Yiclfix + gi|% < (£ 01X + gn)®
# Minimize Y ,|f;x + g;|%

. . . standard SOCP has a; = 2

SOCP-solvable
Other Objective Functions

Unvrestricted product of powers

« Minimize —[[,(f;x+ g;)% forany a; € Q"

Logarithmic Chebychev approximation

« Minimize max-,|log(f;x) —log(g;)|

Why no constraint versions?
< Not SOC-representable

+ Transformation changes objective value (but not solution)

SOCP-solvable

Challenges

Applying “linear” solvers to these forms
+ Recursive detection tree-walk
< Recursive transformation tree-walk
+ Heuristic nonnegativity check for linear expressions

Assessing usefulness . . .

< Results from Jared Erickson’s dissertation:
JaredErickson20120u.northwestern.edu

SOCP-solvable

Survey of Test Problems (1)
12% of 1238 nonlinear problems were SOC-solvable!

< not counting QPs with sum-of-squares objectives
< from Vanderbei’'s CUTE & non-CUTE, and netlib/ampl

A variety of forms detected
% hs064 haS 4-/x1 + 32/x2 + 120/.7(3 < 1
< hs036 minimizes —X;{X,X3

% hs073 has 1.645 \/0.28x12 + 0.19x% + 20.5x2 + 0.62x5 <...

» polak4 is a max of sums of squares
* hs049 minimizes (x; — x,)? + (x3 — 1)? + (x, — D* + (x5 — 1)°

2
< emfl_nonconvex has Ziﬂ(xjk — aik) < Sl_zj

SOCP-solvable

Survey of Test Problems (2)

Counted number of test problems . . .
+ Solvable already by a “linear” solver

< Detected as SOCP-equivalent by our routines

’lﬁl

P

“Linear”’

SOCP-equiv

Who Should Linearize It?

The AMPL user
The AMPL processor
The AMPL-solver interface

The solver

The AMPL User

Advantages
+ Can exploit special knowledge of the problem
< Doesn’t have to be programmed

Disadvantages
< May not know the best way to linearize
<+ May have better ways to use the time
< Can make mistakes

The AMPL Processor

Advantages
< Makes the same linearization available to all solvers
<+ Has a high-level view of the problem

Disadvantages
+ Is a very complicated program
<+ Can’t take advantage of special solver features

The AMPL-Solver Interface

Advantages
<+ Works on simplified problem instances
« Can use same ideas for many solvers, but also
< Can tailor linearization to solver features

Disadvantages

+ Creates an extra layer of complication

Robert Fourer, Strategies for “Not Linear” Optimization
INFORMS Opt Soc Conf — Houston 6-8 March 2014

80




The Solver

Advantages
< Qught to know what’s best for it

+ Can integrate linearization with other activities

Disadvantages
< May not incorporate best practices
< Is complicated enough already

